
Abbiamo parlato di…
 definizione campi e variabili
 definizione del formato
 uso delle costanti data/ora
 reperimento data/ora correnti
 conversione tra tipi dati

 conversione di durate
 somma e sottrazione di durate

 differenza
 comparazione

 controllo
 estrazione porzioni
 arrotondamento

Power
coffee 

29/2021

Tutto (o quasi) su date e ore:
https://www.markonetools.it/tutto-o-quasi-su-date-e-ore/

Power coffee 
32/2021

Power coffee 
33/2021

Power coffee 
35/2021

https://www.markonetools.it/power-coffee-n-29-2021/
https://www.markonetools.it/tutto-o-quasi-su-date-e-ore/
https://www.markonetools.it/power-coffee-n-32-2021/
https://www.markonetools.it/power-coffee-n-33-2021/
https://www.markonetools.it/power-coffee-n-35-2021/


Oggi parliamo di…

 valori di default

 valori nulli

Riepilogo definizioni date/ore

Poster tipi dati:
https://www.markonetools.it/ce-ne-di-tutti-i-tipi/

https://www.markonetools.it/ce-ne-di-tutti-i-tipi/


Date/ore e formati

Power coffee - MK1

6



Valori di default nelle tabelle

 Quando si crea una nuova tabella tramite SQL (create table) si può 
assegnare ad un campo data/ora/timestamp:

 un valore di default iniziale

 un valore che viene automaticamente "rinfrescato" dal DB2 ogni volta che si 
esegue un aggiornamento del record

Power coffee - MK1

7

-- data/ora inserimento record

nome_campo timestamp not null default current_timestamp

-- data/ora aggiornamento record

nome_campo timestamp not null generated by default 

for each row on update as row change timestamp



Valori di default in RPG

 Quando in RPG si definisce una variabile di tipo 
data/ora/timestamp la si può inizializzare tramite la keyword inz

 Data del job: inz(*job)

 Data o ora corrente: inz(*sys)

 Una variabile data/ora/timestamp non esplicitamente inizializzata 
conterrà il valore minimo (*loval) dipendente dal formato della 
variabile

DFTDATE.RPGLE

dcl-s DataDft date;          // 01.01.0001

dcl-s OraDft time;           // 00.00.00

dcl-s DataDMYDft date(*DMY); // 01/01/40



Esporre un campo "vuoto"

 Le variabili e i campi data/ora/timestamp possono contenere solo 
date/ora valide oppure il valore null

 Se non si desidera gestire il valore null, le variabili e campi data/ora 
dovranno almeno contenere il valore minimo, che - dal punto di 
vista applicativo - può essere considerato "assenza di valore"

 Per mostrare un campo vuoto a video quando il valore è impostato 
al default 0001-01-01 si può utilizzare la keyword MAPVAL che 
serve per mappare una data, ora o timestamp su un valore diverso 
durante le operazioni di I/O

A            WSDATA          L  B  8 47DATFMT(*EUR)
A                                      MAPVAL(('01.01.0001' *BLANK))

MAPVAL.RPGLE, MAPVALV.DSPF

https://www.ibm.com/docs/en/i/7.5?topic=80-mapval-map-values-keyword-display-files


Esempio MAPVAL

 La keyword MAPVAL è valida solo per i tipi dati data, ora o 
timestamp.

 Si possono specificare anche più valori da mappare. P.es.
MAPVAL(('0001-01-01' *BLANK) ('0001-01-02' *CUR))

Date/ore e valori nulli nel database

 Con i campi numerici (surrogati del tipo data) solitamente una data 
“nulla” viene memorizzata con valore zero

 se si usa un campo data “vero” esso deve contenere una data 
valida altrimenti errore RNQ0112.

 Per consentire l’inserimento del valore null in un campo del 
database bisogna aggiungere nelle keyword delle DDS ALWNULL
oppure se si usa SQL per creare la tabella non bisogna specificare 
not null

 Quando si ordinano i record per un campo che può contenere 
valori null, il valore null è l’ultimo (ovvero il valore massimo).

https://www.ibm.com/docs/en/i/7.5?topic=80-alwnull-allow-null-value-keywordphysical-files-only


Date/ore e valori nulli

 specifica H: keyword ALWNULL(*USRCTL) per gestire i valori 
nulli

 in RPG il valore nullo è rappresentato dalla costante figurativa 
*null

 BIF RPG: %nullind per testare se un campo è nullo o per 
impostare un campo al valore null

 nella definizione di una variabile si può specificare la keyword 
nullind(var-booleana)
dcl-s Campo1 char(10) nullind(nCampo1);
dcl-s nCampo1 ind;
si può testare o impostare il valore della variabile booleana per 
gestire il valore null della variabile principale

NULL2.RPGLE, NULL3.RPGLE

https://www.ibm.com/docs/en/i/7.5?topic=keywords-alwnullno-inputonly-usrctl
https://www.ibm.com/docs/en/i/7.5?topic=functions-nullind-query-set-null-indicator
https://www.ibm.com/docs/en/i/7.5?topic=keywords-nullindnull-indicator

I programmi di esempio 
di Power Coffee sono 
sviluppati e testati sui 
server di PUB400.COM

IBM i 7.5
Power 9009-41G
Dispositivo processore: EP52
Ram: 50 Gb

Welcome to PUB400.COM - Your public IBM i 7.5 server

by Holger Scherer (rzkh.de)

https://pub400.com/
https://www.linkedin.com/in/holger-scherer-46a27217/


Riferimenti

 E-mail aziendale: marco.riva@ivolution.it

 Blog: www.markonetools.it

 E-mail blog: info@markonetools.it

 Linkedin: www.linkedin.com/in/marcoriva-mk1

 Twitter: @MarcoRiva73

 Facebook: https://www.facebook.com/markonetools/

 YouTube: youtube.com/@markonetools

      //*******************************************
      // ESEMPIO DATA/ORA DEFAULT                *
      //*******************************************

      // (c) MarkOneTools 2022

       ctl-opt copyright('MarkOneTools')
               indent(' ')
               option(*nodebugio: *srcstmt)
               dftactgrp(*no)
               datedit(*YMD);

       dcl-s DataCorrSis date inz(*sys);
       dcl-s DataCorrJob date inz(*job);
       dcl-s OraCorrSis time inz(*sys);
       dcl-s TSCorrSis timestamp(12) inz(*sys);

       dcl-s DataDft date;          // 01.01.0001
       dcl-s OraDft time;           // 00.00.00
       dcl-s DataDMYDft date(*DMY); // 01/01/40

       *inlr = *on;


      ****************************************************************
      * ESEMPIO UTILIZZO KEYWORD MAPVAL
      ****************************************************************

      * www.markonetools.it (c) 2019

       ctl-opt copyright('MarkOneTools')
               indent(' ')
               option(*nodebugio: *srcstmt)
               datedit(*dmy);

       dcl-f MAPVALV workstn;

       dcl-s ForEverTrue ind inz(*on);

       dow ForEverTrue;

         exfmt FMT01;

         select;
           // F1=Data default
           when *INKA;
             WSDATA = d'0001-01-01';
             WSDATC = d'0001-01-02';
           // F2=Data corrente
           when *INKB;
             WSDATA = %date();
             WSDATC = %date();
           // F3=Fine
           when *INKC;
             leave;
           other;
         endsl;
         WSDAT2 = WSDATA;
         WSDAT3 = WSDATC;

       enddo;

       *inlr = *on; 


     A****************************************************************
     A* ESEMPIO UTILIZZO KEYWORD MAPVAL
     A****************************************************************
     A* www.markonetools.it (c) 2019
     A                                      DSPSIZ(27 132 *DS3)
     A          R FMT01
     A                                      CA03
     A                                      CF01
     A                                      CF02
     A                                      CLRL(*ALL)
     A                                  1 26'Esempio utilizzo keyword MAPVAL'
     A                                      DSPATR(HI)
     A                                  7 11'Data con MAPVAL (01.01.0001 --> *B-
     A                                      LANK)'
     A                                      DSPATR(HI)
     A            WSDATA          L  B  7 52DATFMT(*EUR)
     A                                      MAPVAL(('01.01.0001' *BLANK))
     A                                  7 67'Data senza MAPVAL'
     A                                      DSPATR(HI)
     A            WSDAT2          L  B  7 87DATFMT(*EUR)
     A                                  8 11'Data con MAPVAL (02.01.0001 --> *C-
     A                                      UR)'
     A                                      DSPATR(HI)
     A            WSDATC          L  B  8 52DATFMT(*EUR)
     A                                      MAPVAL(('02.01.0001' *CUR))
     A                                  8 67'Data senza MAPVAL'
     A                                      DSPATR(HI)
     A            WSDAT3          L  B  8 87DATFMT(*EUR)
     A                                 23  2'F1=Data default'
     A                                 23 19'F2=Data corrente'
     A                                 23 37'F3=Fine'
     A                                 24 41'(c) Copyright MK1 - www.markonetoo-
     A                                      ls.it'
     A                                      DSPATR(HI) 


      * ESEMPIO VALORI NULL

       ctl-opt copyright('MarkOneTools')
               decedit('0,')
               indent(' ')
               option(*nodebugio: *srcstmt: *showcpy)
               expropts(*resdecpos)
               extbinint(*yes)
               alwnull(*usrctl);

       dcl-f EMPLOYEE keyed rename(employee:epmrec);
       dcl-s msg char(52);
       dcl-s ID_Imp like(empno);

     c     *entry        plist
     c                   parm                    ID_Imp
       chain ID_Imp employee;
       if not %found();
         *inlr = *on;
         return;
       endif;

       if %nullind(PHONENO);
         msg = 'Numero di telefono non esistente';
       else;
         msg = 'Numero di telefono: ' + phoneno;
       endif;
       dsply msg;

       *inlr = *on;


      * ESEMPIO VALORI NULL

       ctl-opt copyright('MarkOneTools')
               decedit('0,')
               indent(' ')
               option(*nodebugio: *srcstmt: *showcpy)
               expropts(*resdecpos)
               extbinint(*yes)
               alwnull(*usrctl);

       dcl-s msg char(52);
       dcl-s Field1 char(5) nullind(nField1);
       dcl-s nField1 ind;

       Field1 = 'ABC';

       // con la variabile booleana nField1 testo se la corrispondente
       //  variabile Field1 è nulla
       if nField1;
         msg = 'Field1 è nullo';
       else;
         msg = 'Field1 vale: ' + Field1;
       endif;
       dsply msg;

       // impostando Field1 a null
       //  di conseguenza la variabile booleana nField1 è *on
       %nullind(Field1) = *on;

       if nField1;
         msg = 'Field1 è nullo';
       else;
         msg = 'Field1 vale: ' + Field1;
       endif;
       dsply msg;

       *inlr = *on;

